
Example nº 8

Tendons definition

CivilFEM Manual of Advanced Examples

Example nº 8 – Table of Contents

- 8 EXAMPLE N° 8: TENDONS DEFINITION..... 1
 - 8.1 AIM 1
 - 8.2 DESCRIPTION OF THE EXAMPLE 1
 - 8.3 RESULTS TO BE OBTAINED 2
 - 8.4 ANALYTIC RESOLUTION..... 2
 - 8.5 CALCULATION LOG 5
 - 8.5.1 Introduction 5
 - 8.5.2 Model generation 6
 - 8.5.3 Analysis 7
 - 8.5.4 Postprocess 7
 - 8.6 RESULTS..... 7
 - 8.7 SUMMARY 9


8 EXAMPLE N° 8: TENDONS DEFINITION

8.1 AIM

The aim of this example is to continue practicing with the Prestressed Concrete Module, introducing a cable with a more complex geometry than in the previous example, formed by three parabolic arcs, tangents to each other. In this example the structure also solved.

8.2 DESCRIPTION OF THE EXAMPLE

The example is shown in the following figure



with the following values for the parameters

Parameter	Value	Units
b	0.60	m
h	1.25	m
e _a	0.20	m
e _c	0.48	m
L	25.00	m

A cable is set, made up of three tangent vertical axis parabolas tangents at the quarters of the span length, with a prestressing force $P_0 = 1000$ kN.

8.3 RESULTS TO BE OBTAINED

Determine the forces and moments due to prestressing actions and verify the following extreme values:

	Maximum		Minimum		Unit
	Value	Location	Value	Location	
Bending moment	200	Support	-480.0	Center	m·kN
Axial force	-	-	-1000.0	Constant	kN
Shear force	0	Supports and center	±108.8	Quarters of light	kN

Prestressing losses will not be considered and the effect of the casing diameter will be ignored.

8.4 ANALYTIC RESOLUTION

The prestressing load has a double effect: it causes punctual reactions at the supports and it introduces distributed, normal and tangential loads throughout the beam.

The loads at the supports are:

$$P_h = P \cdot \cos \alpha = 1000 \text{ kN} \cdot \cos 0 = 1000 \text{ kN}$$

$$P_v = P \cdot \sin \alpha = 1000 \text{ kN} \cdot \sin 0 = 0$$

$$M = P_h \cdot e_a = 1000 \text{ kN} \cdot 0,20 \text{ m} = 200 \text{ kN} \cdot \text{m}$$

and the distributed loads:

$$n = \frac{P}{r}$$

$$t = -\mu \cdot n$$

where n is the normal load and t the tangential load.

$$\frac{1}{r} \approx y''(x) = \begin{cases} -0.017408 \text{ m}^{-1} & 0 \text{ m} \leq x \leq 6.25 \text{ m}; 18.75 \text{ m} \leq x \leq 25.0 \text{ m} \\ 0.017408 \text{ m}^{-1} & 6.25 \text{ m} \leq x \leq 18.75 \text{ m} \end{cases}$$


$$n = \frac{P}{r} = \begin{cases} -17.408 \text{ kN/m} & 0 \text{ m} \leq x \leq 6.25 \text{ m}; 18.75 \text{ m} \leq x \leq 25.0 \text{ m} \\ 17.408 \text{ kN/m} & 6.25 \text{ m} \leq x \leq 18.75 \text{ m} \end{cases}$$

In addition, when friction is not considered, t turns out to be zero.

Assuming that the slope of the tendon is small, the normal loads can be considered vertical.


The law of bending moments obtained from the described loads is:

$$M(x) = \begin{cases} -200 + 8.704 \cdot x^2 & 0 \text{ m} \leq x \leq 6.25 \text{ m} \\ -880 + 217.6 \cdot x - 8.705 \cdot x^2 & 6.25 \text{ m} \leq x \leq 18.75 \text{ m} \\ 5240 - 435.2 \cdot x + 8.705 \cdot x^2 & 18.75 \text{ m} \leq x \leq 25.0 \text{ m} \end{cases} \quad (\text{kN} \cdot \text{m})$$



The shear forces law is:

$$V(x) = \begin{cases} 17.408 \cdot x & 0 \text{ m} \leq x \leq 6.25 \text{ m} \\ 217.6 - 17.408 \cdot x & 6.25 \text{ m} \leq x \leq 18.75 \text{ m} \\ -435.2 + 17.408 \cdot x & 18.75 \text{ m} \leq x \leq 25.0 \text{ m} \end{cases} \quad (\text{kN})$$



The axial forces law is constant of value -1000 kN (compression).

8.5 CALCULATION LOG

8.5.1 Introduction

The description of the example does not specify any material to use. The requested results are independent of the material, because it is an isostatic beam and prestressing loads have been neglected.

In order to prevent CivilFEM to calculate and consider prestressing losses, the parameters needed for its calculation are set to zero (anchorage slip, friction coefficients, etc.). In the case of having more than one tendon, it would have been necessary to set a constant tensioning order (all the tendons tensioned at the same moment) to eliminate the losses by elastic shortening of concrete.

Although the losses will be null, it is necessary to select a code for the calculation. In this case the Spanish EHE code has been chosen.

For the geometric definition of the tendon (in elevation) it is enough to set the slopes at the supports (that are null) and the horizontal slope at the center of the span. In order to define the shape of the parabolas the inflection point can be moved, although in this case the default value is valid, because the inflection point is placed at a quarter of the span length from both ends.

The tendon plan definition is a straight line defined by its initial and ending points.

8.5.2 Model generation

```

FINISH
~CFCLEAR,,1

! Parameters
b = 0.60 ! Section width (m)
a = 1.25 ! section high (m)
L = 25.00 ! beam lenght (m)
ea = 0.20 ! tendon position (m)
ec = 0.48 !
p0 = 1e6 ! In. press. force (N)
p = 1e4 ! Load (N/m)

~CODESEL,,EHE,EHE ! Concrete Code
~UNITS,SI

! Material definition
/PREP7
~CFMP,1,LIB,CONCRETE,EHE,HA-45
~CFMP,2,LIB,PREST,EHE,Y1860S7
~CFMP,2,Prest,A ,,0.0
~CFMP,2,Prest,EPSSr ,,0.0
~CFMP,2,Prest,MU ,,0.0
~CFMP,2,Prest,K ,,0.0
~CFMP,2,Prest,PHI ,,0.0

! Beam properties definition
~CSECDMS,1,REC,b,a,1
~BMSHPRO,1,BEAM,1,1,,,44,1,1,Beam 1

! Nodes
N,1 $ N,11,L/4 $ N,21,L/2 $ N,31,3*L/4 $ N,41,L
FILL,1,11 $ FILL,11,21
FILL,21,31 $ FILL,31,41

! Elements
MAT,1 $ ET,1,BEAM44 $ TYPE,1 $ REAL,1
E,1,2 $ EGEN,40,1,1,2,1


! Boundary conditions
D, 1,UX $ D,1,UY $ D,1,UZ $ D,1,ROTX
D,41,UY $ D,41,UZ

! Support Beam
~SBBMDEF

! Tendon Geometry
~PCEPDEF,1, 1,ea
~PCEPDEF,2,21,-eb
~PCEPDEF,3,41,ea
~PCPPDEF,1, 1,0,0
~PCPPDEF,2,41,0,0
~PCTNDEF,1,2,10e-4,0.001,p0,p0,5e6
~PCTNMDF,1,EADD,1
~PCTNMDF,1,EADD,2
~PCTNMDF,1,EADD,3
~PCTNMDF,1,PADD,1
~PCTNMDF,1,PADD,2

! Losses Calculation
~PCLOSS,0,0,,0

! Apply prestressing losses
~PCPL
    
```



8.5.3 Analysis


```
/SOLU  
SOLVE
```

8.5.4 Postprocess

```
/POST1  
! Point to result load state  
~CFSET,,1  
  
! Plot results  
~PLLSFOR,F,X,-1,  
~PLLSFOR,M,Z,-1,  
~PLLSFOR,F,Y,-1
```

8.6 RESULTS

In the following figures the forces and moments distribution are shown



From them, the following summary table can be extracted:

	Maximum		Minimum		Unit
	Value	Location	Value	Location	
Bending moment	200	Support	-480.00	Center	m·kN
Axial force	-	-	-1000.00	Constant	kN
Shear force	0	Supports and center	±108.02	Quarters of light	kN

As it can be seen, the differences respect to the analytically obtained results are minimum.

8.7 SUMMARY

With this exercise the reader has modelled and solved a simple prestressed concrete structure, in which the tendon, or active reinforcement, is made up by a series of vertical axis parabolas.

The example also serves for the user to become familiar with the concept of prestressing forces and moments and to distinguish between isostatic or primary forces and moments (that are the total forces and moments in this case) and hyperstatic or secondary (zero in this example).